Curves and Surfaces for CAD systems



Curve for Design

---

1) Parametric Curve and its properties

```
In[252]:= ClearAll["Global`*"]
x1 = θ;
y1 = Sin [θ];
MyCurve[θ_] = {x1, y1};
UnitTangent[θ_] := {1/Sqrt[1 + Cos[θ]^2], Cos[θ]/Sqrt[1 + Cos[θ]^2]}
UnitNormal[θ_] := {-Cos[θ]/Sqrt[1 + Cos[θ]^2], 1/Sqrt[1 + Cos[θ]^2]}
UnitTangentDeri1[θ_] :=
  {Sqrt[1 + Cos[θ]^2] Cot[θ] Sqrt[Sin[θ]^2/(1 + Cos[θ]^2)^2], - Sqrt[1 + Cos[θ]^2] Csc[θ] Sqrt[Sin[θ]^2/(1 + Cos[θ]^2)^2]};
CurveElp[θ_] := - Sin[θ]/(1 + Cos[θ]^2)^(3/2)
Quiet[Manipulate[
  Show[
   {ParametricPlot[MyCurve[θ], {θ, -π, π}, PlotRange → All],
    Graphics[{Red, Arrow[{MyCurve[a], MyCurve[a] + UnitTangent[a]}]}],
    Graphics[{Black, Arrow[{MyCurve[a], MyCurve[a] + UnitNormal[a]}]}],
    Graphics[{Green, Arrow[{MyCurve[a], MyCurve[a] + UnitTangentDeri1[a]}]}],
    Graphics[Circle[MyCurve[a] +
       (UnitTangentDeri1[a]) * Abs[1 / CurveElp[a]], Abs[1 / CurveElp[a]]]],
    Graphics[Point[MyCurve[a] + (UnitTangentDeri1[a]) * Abs[1 / CurveElp[a]]]],
    Graphics[Point[MyCurve[a]]]}], {{a, -1.8}, -π + 0.1, π - 0.1}]]
```

Out[260]=



1) Extra: Curve Synthesis

In[261]:=
```
intrinsic[fun_, a_ : 0, {c_ 0, d_ : 0, theta0_ : 0}, optsnd___, {smin_, smax_}][t_] :=
  Flatten[Module[{x, y, theta},
    {x[t], y[t]} /. NDSolve[{x'[ss] == Cos[theta[ss]], y'[ss] == Sin[theta[ss]],
       theta'[ss] = fun[ss], x[a] == c, y[a] == d, theta[a] == theta0},
      {x, y, theta}, {ss, smin, smax}, optsnd]]]
plotintrinsic[fun_, a_ : 0, {c_ : 0, d_ : 0, theta0_ : 0}, optsnd___,
   {smin_, smax_}, optspp___] := ParametricPlot[Module[{x, y, theta},
     {x[t], y[t]} /. NDSolve[{x'[ss] == Cos[theta[ss]], y'[ss] == Sin[theta[ss]],
        theta'[ss] == fun[ss], x[a] == c, y[a] == d, theta[a] == theta0},
       {x, y, theta}, {ss, smin, smax}, optsnd, MaxSteps → 1000]] // Evaluate,
   {t, smin, smax}, PlotRange → All, AspectRatio → Automatic,
   PlotStyle → Directive[Black, Thick] , optspp];
```

Curvature for case [s] = a s

In[27]:=
```
Quiet[Manipulate[{plotintrinsic[# * a &, 0, {0, 0, 0}, {-5, 5}, PlotPoints → 80],
   Plot[ s * a, {s, -5, 5}, PlotRange → All]}, {a, -10, 10}]]
```

Out[27]=

a

```
{plotintrinsic[#1 FE`a$$249 &, 0, {0, 0, 0}, {-5, 5}, PlotPoints → 80],
```



}

Curvature for case [s] = a*Sin[s]

In[28]:=
```
Quiet[Manipulate[{plotintrinsic[Sin[#] * a &, 0, {0, 0, 0}, {-5, 5}, PlotPoints → 80],
   Plot[ Sin[s] * a, {s, -5, 5}, PlotRange → All]}, {a, -10, 10}]]
```

Out[28]=

a

```
{plotintrinsic[Sin[#1] FE`a$$250 &, 0, {0, 0, 0}, {-5, 5}, PlotPoints → 80],
```



}

Bezier Curves

1) Linear Bezier

Define the linear Bezier equation.

In[29]:=
```
B1[t_, P0_, P1_] := (1 - t) P0 + t * P1
```

Draw a linear Bezier line connecting (0,0) and (2,1). Show the curve and it's control points.

In[30]:=
```
P1 = {p10, p11} = {{0, 0}, {2, 1}};
f11 = ParametricPlot[B1[t, p10, p11], {t, 0, 1}];
f12 = Graphics[{Red, PointSize[Large], Point[P1]}];
Show[f12, f11]
```

Out[33]=



In[34]:=
```
DynamicModule[
 {P0 = {-2, -2}, P1 = {2, 2}},
 LocatorPane[
  Dynamic[{P0, P1}],
  Dynamic[{
    Show[{
      ParametricPlot[(1 - t) P0 + t * P1, {t, 0, 1},
       PlotRange → {{-5, 5}, {-5, 5}}, Axes → False, Frame → True],
      ListPlot[{P0, P1}, Joined → True, PlotStyle → Red]
     }],
    Dynamic[{P0, P1}]}
  ]]]
```

Out[34]= {  , {{-2, -2}, {2, 2}}}

2) Quadratic Bezier

Define the quadratic Bezier equation.

In[35]:=
```
B2[t_, p0_, p1_, p2_] := p0 (1 - t)^2 + 2 t (1 - t) p1 + t^2 p2
basisFunction = {b0[t_] = (1 - t)^2, b1[t_] = 2 t (1 - t), b2[t_] = t^2};
```

Basis Function plot

In[37]:=
```
Plot[Evaluate[Table[basisFunction[[n]], {n, 0, 3}]],
 {t, 0, 1}, PlotRange → All, AxesLabel → {"parameter t", "b_i" }]
```

Out[37]=



Draw a quadratic Bezier with control points (0,0), (2,3), and (4,-1). Draw the control points and polygons, then show the curve, control points and polygons in a single plot.

In[38]:=
```
P2 = {p20, p21, p22} = {{0, 0}, {2, 3}, {4, -1}};
f21 = ParametricPlot[B2[t, p20, p21, p22], {t, 0, 1}];
f22 = Graphics[{Line[P2], Red, PointSize[Large], Point[P2]}];
Show[f22, f21]
```

Out[41]=



3) Interactive Bezier Cubic

In[42]:=
```
DynamicModule[
 {P0 = {0, 0}, P1 = {1, 2}, P2 = {3, 0}, P3 = {5, 2}},
 LocatorPane[
  Dynamic[{P0, P1, P2, P3}],
  Dynamic[{
    Show[{
      ParametricPlot[(1 - t)^3 P0 + 3 t (1 - t)^2 * P1 + 3 t^2 (1 - t) * P2 + t^3 P3, {t, 0, 1},
       PlotRange → {{-5, 5}, {-5, 5}}, Axes → False, Frame → True],
      ListPlot[{P0, P1, P2, P3}, Joined → True, PlotStyle → Red]
     }],
    Dynamic[{P0, P1, P2, P3}]}
  ]]]
```

Out[42]= {, {{-2.785, -1.4}, {1, 2}, {3.47, -1.22}, {3.89, 2.4}}}

4) de Casteljau Algo

In[551]:=
```
ClearAll["Global`*"]
deCasteljau[t_, P0_, P1_] := (1 - t) P0 + t P1
{p0, p1, p2, p3} = {{0, 0}, {1, 2}, {3, 0}, {5, 2}};
Manipulate[Show[{ListPlot[{p0, p1, p2, p3}, Joined → True, PlotStyle → Red],
   ListPlot[{p10 = (1 - t) p0 + t p1,
     p11 = (1 - t) p1 + t p2, p12 = (1 - t) p2 + t p3}, PlotStyle → Black],
   ListPlot[{p10, p11, p12}, Joined → True, PlotStyle → Blue],
   ListPlot[{p20 = (1 - t) p10 + t p11, p21 = (1 - t) p11 + t p12}, PlotStyle → Black],
   ListPlot[{p20, p21}, Joined → True, PlotStyle → Green],
   ListPlot[{p30 = (1 - t) p20 + t p21}, PlotStyle → Red],
   ParametricPlot[{(1 - t)^3 p0 + 3 t (1 - t)^2 * p1 + 3 t^2 (1 - t) * p2 + t^3 p3}, {t, 0, 1}]
  }, Frame → True, PlotRange → All], {t, 0, 1}]
```

Out[554]=



5) Rational Bezier

In[555]:=
```
ClearAll["Global`*"]
Bez[t_, w0_, w1_, w2_, w3_] :=
  (w0 * (1 - t)^3 P0 + w1 * 3 t (1 - t)^2 P1 + w2 * 3 t^2 (1 - t) P2 + w3 * t^3 P3) /
  (w0 * (1 - t)^3 + w1 * 3 t (1 - t)^2 + w2 * 3 t^2 (1 - t) + w3 * t^3)
{P0, P1, P2, P3} = {{0, 3}, {1, 0}, {5, 5}, {5, 0}};
Bez[t, w0, w1, w2, w3]
Manipulate[Show[{ParametricPlot[Bez[t, w0, w1, w2, w3], {t, 0, 1}],
    ListPlot[{{P0, P1, P2, P3}}, Joined → True, PlotStyle → Gray],
    ListPlot[{{P0, P1, P2, P3}}, PlotStyle → Red], Graphics[{Text[P₀, P0 - 0.2]}]},
   PlotRange → {{-1, 6}, {-1, 6}}, AxesOrigin → {0, 0}],
 {w0, 0.1, 1}, {w1, 0.1, 1}, {w2, 0.1, 1}, {w3, 0.1, 1}]
```

Out[558]=

$$\left\{\frac{3\,(1-t)^2\,t\,w1+15\,(1-t)\,t^2\,w2+5\,t^3\,w3}{(1-t)^3\,w0+3\,(1-t)^2\,t\,w1+3\,(1-t)\,t^2\,w2+t^3\,w3},\right.$$
$$\left.\frac{3\,(1-t)^3\,w0+15\,(1-t)\,t^2\,w2}{(1-t)^3\,w0+3\,(1-t)^2\,t\,w1+3\,(1-t)\,t^2\,w2+t^3\,w3}\right\}$$

Out[559]=



6) General Bezier: basis functions

In[52]:=
```
Table[Plot[Evaluate[Table[BernsteinBasis[m, n, t], {n, 0, m}]],
  {t, 0, 1}, PlotStyle → Red, PlotRange → All], {m, 1, 9}]
```

Out[52]=



In[560]:=
```
pts = {{0, 0}, {1, 1}, {2, -1}, {3, 0}, {5, 2}, {6, -1}, {7, 3}};
Pygn = Graphics[{Green, Line[pts], Red, Point[pts]}];
BezStandard = Graphics[
   {Red, BezierCurve[pts, SplineDegree → 7], Green, Line[pts], Red, Point[pts]}];

Show[Pygn, BezStandard, PlotRange -> All]
```

Out[563]=



7) Application: Font Design / Morphing

Define the outline of a glyph:

In[564]:=
```
curves = Graphics[{
    Blue, BezierCurve[pts1 = {{2, 3}, {0.8125, 0.625}}],
    Green, BezierCurve[pts2 = {{0.8125, 0.625}, {0.6875, 0.375}, {0.375, 0.25}}],
    Magenta, BezierCurve[pts3 = {{0.375, 0.25}, {1.125, 0.25}}],
    Orange, BezierCurve[pts4 = {{1.125, 0.25}, {0.8125, 0.375}, {0.9375, 0.625}}],
    Black, BezierCurve[{{0.9375, 0.625}, {1.3125, 1.375}}],
    BezierCurve[{{1.3125, 1.375}, {2.4375, 1.375}}],
    BezierCurve[{{2.4375, 1.375}, {2.8125, 0.625}}],
    BezierCurve[{{2.8125, 0.625}, {2.9375, 0.375}, {2.625, 0.25}}],
    BezierCurve[{{2.625, 0.25}, {3.625, 0.25}}],
    BezierCurve[{{3.625, 0.25}, {3.3125, 0.375}, {3.1875, 0.625}}],
    BezierCurve[{{3.1875, 0.625}, {2, 3}}],
    BezierCurve[{{1.875, 2.5}, {1.375, 1.5}}],
    BezierCurve[{{1.375, 1.5}, {2.375, 1.5}}],
    BezierCurve[{{2.375, 1.5}, {1.875, 2.5}}]}];
points = Graphics[{Blue, Point[pts1], Green,
    Point[pts2], Magenta, Point[pts3], Orange, Point[pts4]}];
Show[{curves, points}]
```

Out[566]=



Linear transition from one Bézier curve to another:

In[567]:=
```
pts1 = {{0, -1}, {2, 1}, {4, 2}, {6, 2}};
pts2 = {{2, -1}, {3, 1}, {4, -1}, {6, 0}};
g[p1_, p2_, t_] := (1 - t) p1 + t p2
Animate[Graphics[{Blue, BezierCurve[pts1], Blue,
   BezierCurve[pts2], Thick, Red, BezierCurve[g[pts1, pts2, t]]}],
 {t, 0, 1}, AnimationDirection → ForwardBackward, SaveDefinitions → True]
```

Out[570]=



8) Cubic Bezier spline with G1 continuity

In[571]:=
```
B3[t_, p0_, p1_, p2_, p3_] := p0 (1 - t)^3 + 3 t (1 - t)^2 p1 + 3 t^2 (1 - t) p2 + t^3 p3
P3 = {p30, p31, p32, p33} = {{0, 0}, {2, 3}, {4, -1}, {5, 2}};
P4 = {p40, p41, p42, p43} = {{5, 2}, {5.3, 2.9}, {3, 3}, {3, 1.5}};
f31 = ParametricPlot[B3[t, p30, p31, p32, p33], {t, 0, 1}];
f32 = Graphics[{Line[P3], Red, PointSize[Large], Point[P3]}];
Show[f32, f31]
```

Out[576]=



In[577]:=
```
f42 = Graphics[{Line[P4], Red, PointSize[Large], Point[P4]}];
f41 = ParametricPlot[B3[t, p40, p41, p42, p43], {t, 0, 1}];
Show[f32, f31, f41, f42]
```

Out[579]=



BSplines

1) BSplines degree 2

In[580]:=
```
ClearAll["Global`*"]
```

In[581]:=
```
d = 2;
controlPoints = {b0, b1, b2, b3};
m = (Dimensions[controlPoints][[1]] - 1) + d + 1
e = (m - d);
{t_d, t_e}
```

Out[583]= 6

Out[585]= $\{t_2, t_4\}$

In[586]:=
```
knots = {1, 1, 1, 4, 5, 5, 5};
b0 = {0, 0};
b1 = {1, 1};
b2 = {2, 1};
b3 = {3, 0};
td = knots[[d + 1]];
tmd = knots[[m - d + 1]];
Print[{t_d, t_e}, "=", {td, tmd}]
```

$\{t_2, t_4\}=\{1, 5\}$

In[594]:=
```
k = 0;
Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
   "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 1}] // TableForm
```

Out[595]//TableForm=

$N_{0,0}(x)$	0
$N_{1,0}(x)$	0
$N_{2,0}(x)$	$= \left(\begin{cases} 1 & 1 \le x \le 4 \\ 0 & \text{True} \end{cases} \right)$
$N_{3,0}(x)$	$= \left(\begin{cases} 1 & 4 \le x \le 5 \\ 0 & \text{True} \end{cases} \right)$
$N_{4,0}(x)$	0
$N_{5,0}(x)$	0

In[91]:=
```
k = 1;
Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
   "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 2}] // TableForm
Plot[Evaluate[Table[BSplineBasis[{k, knots}, i, x], {i, 0, m - 2}]],
 {x, 1, 5}, AspectRatio → Automatic, PlotRange → All]
```

Out[92]//TableForm=

$N_{0,1}(x)$ 0

$N_{1,1}(x)$ = $\left(\begin{cases} \frac{4-x}{3} & 1 \le x \le 4 \\ 0 & \text{True} \end{cases}\right)$

$N_{2,1}(x)$ = $\left(\begin{cases} 5-x & 4 \le x \le 5 \\ \frac{1}{3}(-1+x) & 1 \le x < 4 \\ 0 & \text{True} \end{cases}\right)$

$N_{3,1}(x)$ = $\left(\begin{cases} -4+x & 4 \le x \le 5 \\ 0 & \text{True} \end{cases}\right)$

$N_{4,1}(x)$ 0

Out[93]=



In[94]:=
```
k = 2;
Print[{t_d, t_e}, "=", {td, tmd}]
BSpline = Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
    "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 3}] // TableForm
Plot[Evaluate[Table[BSplineBasis[{k, knots}, i, x], {i, 0, m - 3}]],
 {x, 1, 5}, AspectRatio → Automatic, PlotRange → All]
```

$\{t_2, t_4\}=\{1, 5\}$

Out[96]//TableForm=

$$N_{0,2}(x) = \left(\begin{cases} \frac{1}{9}\left(16-8x+x^2\right) & 1 \le x \le 4 \\ 0 & \text{True} \end{cases}\right)$$

$$N_{1,2}(x) = \left(\begin{cases} \frac{1}{36}\left(-31+38x-7x^2\right) & 1 \le x < 4 \\ \frac{1}{4}\left(25-10x+x^2\right) & 4 \le x \le 5 \\ 0 & \text{True} \end{cases}\right)$$

$$N_{2,2}(x) = \left(\begin{cases} \frac{1}{4}\left(-85+42x-5x^2\right) & 4 \le x \le 5 \\ \frac{1}{12}\left(1-2x+x^2\right) & 1 \le x < 4 \\ 0 & \text{True} \end{cases}\right)$$

$$N_{3,2}(x) = \left(\begin{cases} 16-8x+x^2 & 4 \le x \le 5 \\ 0 & \text{True} \end{cases}\right)$$

Out[97]=


In[98]:=
```
Piece1 = ParametricPlot[1/9 (16 - 8 x + x^2) b0 + 1/36 (-31 + 38 x - 7 x^2) b1 + 1/12 (1 - 2 x + x^2) b2,
   {x, 1, 4}, Frame → True, Axes → False, PlotRange → All, PlotStyle → Green];
```

In[99]:=
```
Piece2 = ParametricPlot[1/4 (25 - 10 x + x^2) b1 + 1/4 (-85 + 42 x - 5 x^2) b2 + (16 - 8 x + x^2) b3,
   {x, 4, 5}, Frame → True, Axes → False, PlotRange → All, PlotStyle → Magenta];
ctlpts = ListLinePlot[{b0, b1, b2, b3}, Mesh → Full, PlotStyle → Gray];
Show[{Piece1, Piece2, ctlpts}, AspectRatio → Automatic]
```

Out[101]=


2) BSplines degree 2

In[102]:=
```
ClearAll["Global`*"]
```

In[103]:=
```
d = 2;
controlPoints = {b0, b1, b2, b3, b4};
m = (Dimensions[controlPoints][[1]] - 1) + d + 1
e = (m - d);
{t_d, t_e}
```

Out[105]= 7

Out[107]= $\{t_2, t_5\}$

In[108]:=
```
knots = {1, 1, 1, 2, 4, 5, 5, 5};
b0 = {0, 0};
b1 = {1, 1};
b2 = {2, 1};
b3 = {3, 0};
b4 = {4, 3};
td = knots[[d + 1]];
tmd = knots[[m - d + 1]];
Print[{t_d, t_e}, "=", {td, tmd}]
```

$\{t_2, t_5\}=\{1, 5\}$

In[117]:=
```
k = 0;
Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
   "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 1}] // TableForm
```

Out[118]//TableForm=

$N_{0,0}(x)$	0
$N_{1,0}(x)$	0
$N_{2,0}(x)$	$= \left(\begin{cases} 1 & 1 \le x \le 2 \\ 0 & \text{True} \end{cases}\right)$
$N_{3,0}(x)$	$= \left(\begin{cases} 1 & 2 \le x \le 4 \\ 0 & \text{True} \end{cases}\right)$
$N_{4,0}(x)$	$= \left(\begin{cases} 1 & 4 \le x \le 5 \\ 0 & \text{True} \end{cases}\right)$
$N_{5,0}(x)$	0
$N_{6,0}(x)$	0

In[119]:=
```
k = 1;
Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
   "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 2}] // TableForm
Plot[Evaluate[Table[BSplineBasis[{k, knots}, i, x], {i, 0, m - 2}]],
 {x, 1, 5}, AspectRatio → Automatic, PlotRange → All]
```

Out[120]//TableForm=

$N_{0,1}(x) \quad 0$

$N_{1,1}(x) \quad = \left(\begin{cases} 2-x & 1 \le x \le 2 \\ 0 & \text{True} \end{cases} \right)$

$N_{2,1}(x) \quad = \left(\begin{cases} \frac{4-x}{2} & 2 \le x \le 4 \\ -1+x & 1 \le x < 2 \\ 0 & \text{True} \end{cases} \right)$

$N_{3,1}(x) \quad = \left(\begin{cases} 5-x & 4 \le x \le 5 \\ \frac{1}{2}(-2+x) & 2 \le x < 4 \\ 0 & \text{True} \end{cases} \right)$

$N_{4,1}(x) \quad = \left(\begin{cases} -4+x & 4 \le x \le 5 \\ 0 & \text{True} \end{cases} \right)$

$N_{5,1}(x) \quad 0$

Out[121]=



In[122]:=
```
k = 2;
Print[{t_d, t_e}, "=", {td, tmd}]
BSpline = Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
    "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 3}] // TableForm
Plot[Evaluate[Table[BSplineBasis[{k, knots}, i, x], {i, 0, m - 3}]],
 {x, 1, 5}, AspectRatio → Automatic, PlotRange → All]
```

$\{t_2, t_5\} = \{1, 5\}$

Out[124]//TableForm=

$$N_{0,2}(x) = \begin{cases} 4 - 4x + x^2 & 1 \le x \le 2 \\ 0 & \text{True} \end{cases}$$

$$N_{1,2}(x) = \begin{cases} \frac{1}{6}(16 - 8x + x^2) & 2 \le x \le 4 \\ -\frac{2}{3}(5 - 7x + 2x^2) & 1 \le x < 2 \\ 0 & \text{True} \end{cases}$$

$$N_{2,2}(x) = \begin{cases} \frac{1}{3}(-7 + 6x - x^2) & 2 \le x < 4 \\ \frac{1}{3}(25 - 10x + x^2) & 4 \le x \le 5 \\ \frac{1}{3}(1 - 2x + x^2) & 1 \le x < 2 \\ 0 & \text{True} \end{cases}$$

$$N_{3,2}(x) = \begin{cases} \frac{1}{6}(4 - 4x + x^2) & 2 \le x < 4 \\ -\frac{2}{3}(35 - 17x + 2x^2) & 4 \le x \le 5 \\ 0 & \text{True} \end{cases}$$

$$N_{4,2}(x) = \begin{cases} 16 - 8x + x^2 & 4 \le x \le 5 \\ 0 & \text{True} \end{cases}$$

Out[125]=



In[126]:=
```
Piece1 = ParametricPlot[(4 - 4 x + x^2) b0 - 2/3 (5 - 7 x + 2 x^2) b1 + 1/3 (1 - 2 x + x^2) b2,
   {x, 1, 2}, Frame → True, Axes → False, PlotRange → All, PlotStyle → Green];

Piece2 = ParametricPlot[1/6 (16 - 8 x + x^2) b1 + 1/3 (-7 + 6 x - x^2) b2 + 1/6 (4 - 4 x + x^2) b3,
   {x, 2, 4}, Frame → True, Axes → False, PlotRange → All, PlotStyle → Red];

Piece3 = ParametricPlot[1/3 (25 - 10 x + x^2) b2 - 2/3 (35 - 17 x + 2 x^2) b3 + (16 - 8 x + x^2) b4,
   {x, 4, 5}, Frame → True, Axes → False, PlotRange → All, PlotStyle → Magenta];
```

In[129]:=
```
ctlpts = ListLinePlot[{b0, b1, b2, b3, b4}, Mesh → Full, PlotStyle → Gray];
Show[{Piece1, Piece2, Piece3, ctlpts}, AspectRatio → Automatic]
```

Out[130]=



3) BSplines degree 3

In[131]:=
```
ClearAll["Global`*"]
```

In[132]:=
```
d = 3;
controlPoints = {b0, b1, b2, b3, b4};
m = (Dimensions[controlPoints][[1]] - 1) + d + 1
e = (m - d);
{t_d, t_e}
```

Out[134]= 8

Out[136]= $\{t_3, t_5\}$

In[137]:=
```
knots = {1, 1, 1, 1, 3, 4, 5, 6, 7};
b0 = {0, 0};
b1 = {1, 1};
b2 = {2, 1};
b3 = {3, 0};
b4 = {4, 3};
td = knots[[d + 1]];
tmd = knots[[m - d + 1]];
Print[{t_d, t_e}, "=", {td, tmd}]
```

$\{t_3, t_5\}=\{1, 4\}$

In[146]:=
```
k = 0;
Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
   "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 1}] // TableForm
```

Out[147]//TableForm=

$N_{0,0}(x) \quad 0$

$N_{1,0}(x) \quad 0$

$N_{2,0}(x) \quad 0$

$N_{3,0}(x) = \left(\begin{cases} 1 & 1 \le x \le 3 \\ 0 & \text{True} \end{cases}\right)$

$N_{4,0}(x) = \left(\begin{cases} 1 & 3 \le x \le 4 \\ 0 & \text{True} \end{cases}\right)$

$N_{5,0}(x) = \left(\begin{cases} 1 & 4 \le x \le 5 \\ 0 & \text{True} \end{cases}\right)$

$N_{6,0}(x) = \left(\begin{cases} 1 & 5 \le x \le 6 \\ 0 & \text{True} \end{cases}\right)$

$N_{7,0}(x) = \left(\begin{cases} 1 & 6 \le x \le 7 \\ 0 & \text{True} \end{cases}\right)$

In[148]:=
```
k = 1;
Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
   "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 2}] // TableForm
Plot[Evaluate[Table[BSplineBasis[{k, knots}, i, x], {i, 0, m - 2}]],
 {x, 1, 7}, AspectRatio → Automatic, PlotRange → All]
```

Out[149]//TableForm=

$N_{0,1}(x) \quad 0$

$N_{1,1}(x) \quad 0$

$N_{2,1}(x) = \left(\begin{cases} \frac{3-x}{2} & 1 \le x \le 3 \\ 0 & \text{True} \end{cases}\right)$

$N_{3,1}(x) = \left(\begin{cases} 4 - x & 3 \le x \le 4 \\ \frac{1}{2}(-1 + x) & 1 \le x < 3 \\ 0 & \text{True} \end{cases}\right)$

$N_{4,1}(x) = \left(\begin{cases} 5 - x & 4 \le x \le 5 \\ -3 + x & 3 \le x < 4 \\ 0 & \text{True} \end{cases}\right)$

$N_{5,1}(x) = \left(\begin{cases} 6 - x & 5 \le x \le 6 \\ -4 + x & 4 \le x < 5 \\ 0 & \text{True} \end{cases}\right)$

$N_{6,1}(x) = \left(\begin{cases} 7 - x & 6 \le x \le 7 \\ -5 + x & 5 \le x < 6 \\ 0 & \text{True} \end{cases}\right)$

Out[150]=



In[151]:=
```
k = 2;
BSpline = Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
    "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 3}] // TableForm
Plot[Evaluate[Table[BSplineBasis[{k, knots}, i, x], {i, 0, m - 3}]],
 {x, 1, 7}, AspectRatio → Automatic, PlotRange → All]
```

Out[152]//TableForm=

$N_{0,2}(x)$ $\quad 0$

$$N_{1,2}(x) = \begin{cases} \frac{1}{4}\left(9-6x+x^2\right) & 1 \le x \le 3 \\ 0 & \text{True} \end{cases}$$

$$N_{2,2}(x) = \begin{cases} \frac{1}{12}\left(-17+22x-5x^2\right) & 1 \le x < 3 \\ \frac{1}{3}\left(16-8x+x^2\right) & 3 \le x \le 4 \\ 0 & \text{True} \end{cases}$$

$$N_{3,2}(x) = \begin{cases} \frac{1}{6}\left(-53+34x-5x^2\right) & 3 \le x < 4 \\ \frac{1}{2}\left(25-10x+x^2\right) & 4 \le x \le 5 \\ \frac{1}{6}\left(1-2x+x^2\right) & 1 \le x < 3 \\ 0 & \text{True} \end{cases}$$

$$N_{4,2}(x) = \begin{cases} \frac{1}{2}\left(-39+18x-2x^2\right) & 4 \le x < 5 \\ \frac{1}{2}\left(36-12x+x^2\right) & 5 \le x \le 6 \\ \frac{1}{2}\left(9-6x+x^2\right) & 3 \le x < 4 \\ 0 & \text{True} \end{cases}$$

$$N_{5,2}(x) = \begin{cases} \frac{1}{2}\left(-59+22x-2x^2\right) & 5 \le x < 6 \\ \frac{1}{2}\left(49-14x+x^2\right) & 6 \le x \le 7 \\ \frac{1}{2}\left(16-8x+x^2\right) & 4 \le x < 5 \\ 0 & \text{True} \end{cases}$$

Out[153]=



In[154]:=
```
k = 3;
Print[{t_d, t_e}, "=", {td, tmd}]
BSpline = Table[{TraditionalForm[BSplineBasis[{k, knots}, i, x]],
    "=" PiecewiseExpand[BSplineBasis[{k, knots}, i, x]]}, {i, 0, m - 4}] // TableForm
Plot[Evaluate[Table[BSplineBasis[{k, knots}, i, x], {i, 0, m - 4}]],
 {x, 1, 7}, AspectRatio → Automatic, PlotRange → All]
```

$\{t_3, t_5\} = \{1, 4\}$

Out[156]//TableForm=

$$N_{0,3}(x) = \begin{cases} \frac{1}{8}\left(27 - 27x + 9x^2 - x^3\right) & 1 \le x \le 3 \\ 0 & \text{True} \end{cases}$$

$$N_{1,3}(x) = \begin{cases} \frac{1}{9}\left(64 - 48x + 12x^2 - x^3\right) & 3 \le x \le 4 \\ \frac{1}{72}\left(-217 + 345x - 147x^2 + 19x^3\right) & 1 \le x < 3 \\ 0 & \text{True} \end{cases}$$

$$N_{2,3}(x) = \begin{cases} \frac{1}{72}\left(49 - 111x + 75x^2 - 13x^3\right) & 1 \le x < 3 \\ \frac{1}{8}\left(125 - 75x + 15x^2 - x^3\right) & 4 \le x \le 5 \\ \frac{1}{72}\left(-923 + 861x - 249x^2 + 23x^3\right) & 3 \le x < 4 \\ 0 & \text{True} \end{cases}$$

$$N_{3,3}(x) = \begin{cases} \frac{1}{24}\left(269 - 267x + 87x^2 - 9x^3\right) & 3 \le x < 4 \\ \frac{1}{6}\left(216 - 108x + 18x^2 - x^3\right) & 5 \le x \le 6 \\ \frac{1}{24}\left(-1 + 3x - 3x^2 + x^3\right) & 1 \le x < 3 \\ \frac{1}{24}\left(-1011 + 693x - 153x^2 + 11x^3\right) & 4 \le x < 5 \\ 0 & \text{True} \end{cases}$$

$$N_{4,3}(x) = \begin{cases} \frac{1}{6}\left(229 - 165x + 39x^2 - 3x^3\right) & 4 \le x < 5 \\ \frac{1}{6}\left(343 - 147x + 21x^2 - x^3\right) & 6 \le x \le 7 \\ \frac{1}{6}\left(-27 + 27x - 9x^2 + x^3\right) & 3 \le x < 4 \\ \frac{1}{6}\left(-521 + 285x - 51x^2 + 3x^3\right) & 5 \le x < 6 \\ 0 & \text{True} \end{cases}$$

Out[157]=



In[158]:=
```
Piece1 = ParametricPlot[1/8 (27 - 27 x + 9 x^2 - x^3) b0 + 1/72 (-217 + 345 x - 147 x^2 + 19 x^3) b1 +
    1/72 (49 - 111 x + 75 x^2 - 13 x^3) b2 + 1/24 (-1 + 3 x - 3 x^2 + x^3) b3, {x, 1, 3},
   Frame → True, Axes → False, PlotRange → All, PlotStyle → Green];
Piece2 = ParametricPlot[1/9 (64 - 48 x + 12 x^2 - x^3) b1 + 1/72 (-923 + 861 x - 249 x^2 + 23 x^3) b2 +
    1/24 (269 - 267 x + 87 x^2 - 9 x^3) b3 + 1/6 (-27 + 27 x - 9 x^2 + x^3) b4,
   {x, 3, 4}, Frame → True, Axes → False, PlotRange → All, PlotStyle → Red];
```

In[160]:=
```
ctlpts = ListLinePlot[{b0, b1, b2, b3, b4}, Mesh → Full, PlotStyle → Gray];
Show[{Piece1, Piece2, ctlpts}, AspectRatio → Automatic]
```

Out[161]=



4) Built-in BSpline

Degree

In[163]:=
```
pts = {{0, 0}, {1, 1}, {2, -1}, {3, 0}, {4, -2}, {5, 1}};
Table[Graphics[{Green, Line[pts], Red, Point[pts],
   Black, BSplineCurve[pts, SplineDegree → d]}], {d, 1, 4}]
```

Out[164]=



Knot vectors

In[165]:=
```
pts = {{0, 0}, {0, 2}, {2, 3}, {4, 0}, {6, 3}, {8, 2}, {8, 0}};
Graphics[{Green, Line[pts], Red, Point[pts], Black,
  BSplineCurve[pts, SplineKnots → {0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10}]}]
```

Out[166]=



Using SplineKnots as automatic:

In[167]:=
```
pts = {{0, 0}, {0, 2}, {2, 3}, {4, 0}, {6, 3}, {8, 2}, {8, 0}};
Graphics[{Green, Line[pts], Red, Point[pts],
  Black, BSplineCurve[pts, SplineKnots → Automatic]}]
```

Out[168]=



Defining SplineKnots as automatic with d=3,n=6,, m=d+n+1=10:

In[169]:=
```
pts = {{0, 0}, {0, 2}, {2, 3}, {4, 0}, {6, 3}, {8, 2}, {8, 0}};
Graphics[{Green, Line[pts], Red, Point[pts], Black,
  BSplineCurve[pts, SplineKnots → {0, 0, 0, 0, 1, 1, 1, 2, 2, 2, 2}]}]
```

Out[170]=



SplineClosed

In[171]:=
```
pts = {{0, 0}, {0, 1}, {1, 1}, {1, 0}};
Graphics[{Green, Line[pts], Red, Point[pts], Black, BSplineCurve[pts]}]
```

Out[172]=



Smoothly closed B-spline curve with the same control points:

In[173]:=
```
Graphics[
 {Green, Line[pts], Red, Point[pts], Black, BSplineCurve[pts, SplineClosed → True]}]
```

Out[173]=



NURBS

Perfect circle using NURBS:

In[174]:=
```
pts = {{.5, 0}, {1, 0}, {1, 1}, {.5, 1}, {0, 1}, {0, 0}, {.5, 0}};
w = {1, .5, .5, 1, .5, .5, 1};
k = {0, 0, 0, 1 / 4, 1 / 2, 1 / 2, 3 / 4, 1, 1, 1};
Graphics[{Green, Line[pts], Red, Point[pts],
  Black, BSplineCurve[pts, SplineWeights → w, SplineKnots → k]}]
```

Out[177]=



Surfaces

1) Explicit Surfaces

In[178]:=
```
ClearAll["Global`*"]
Plot3D[Sin[x + y^2], {x, -3, 3}, {y, -2, 2}, MeshStyle → Gray, Mesh → 5]
```

Out[179]=



2) Implicit Surfaces

In[180]:=
```
ClearAll["Global`*"]
iBall[x_, y_, z_] = x^2 + y^2 + z^2 - 10;
f1 = ContourPlot3D[iBall[x, y, z] == 0,
   {x, -6, 6}, {y, -6, 6}, {z, -6, 6}, MeshStyle → Gray, Mesh → 3,
   ContourStyle → Opacity[0.5], ColorFunction → "Pastel"];
{iBall[1, 1, 2 √2], iBall[5, 5, 5], iBall[1, 1, -1], iBall[1, 1, -5]}
f2 = Graphics3D[{PointSize[Large], Red, Point[{1, 1, 2 √2}]}];
f3 = Graphics3D[{PointSize[Large], Magenta, Point[{5, 5, 5}]}];
f4 = Graphics3D[{PointSize[Large], Green, Point[{1, 1, -1}]}];
f5 = Graphics3D[{PointSize[Large], Blue, Point[{1, 5, -5}]}];
Show[{f1, f2, f3, f4, f5}]
```

Out[183]= {0, 65, -7, 17}

Out[188]=



3) Parametric Surface

a) Bilinear Patch

In[190]:=
```
ClearAll["Global`*"]
p00 = {1, 1, 2 √2};
p01 = {5, 5, 5};
p10 = {1, 1, -1};
p11 = {1, 5, -5};
fp00 = Graphics3D[{PointSize[Large], Red, Point[p00]}];
fp01 = Graphics3D[{PointSize[Large], Magenta, Point[p01]}];
fp10 = Graphics3D[{PointSize[Large], Green, Point[p10]}];
fp11 = Graphics3D[{PointSize[Large], Blue, Point[p11]}];
```

In[199]:=
```
Bilinear[u_, v_] = (1 - u) * (1 - v) p00 + (1 - u) v p01 + u (1 - v) p10 + u v p11
fB = ParametricPlot3D[Bilinear[u, v], {u, 0, 1}, {v, 0, 1}];
Show[{fB, fp00, fp01, fp10, fp11}]
```

Out[199]= $\{(1-u)\ (1-v)+u\ (1-v)+5\ (1-u)\ v+u\ v,\ (1-u)\ (1-v)+u\ (1-v)+5\ (1-u)\ v+5\ u\ v,\ 2\sqrt{2}\ (1-u)\ (1-v)-u\ (1-v)+5\ (1-u)\ v-5\ u\ v\}$

Out[201]=



b) Lofting

In[202]:=
```
ClearAll["Global`*"]
C1[u_] := {Sin[u], Cos[u]}
C2[u_] := {(2 - 0.1 u) Sin[u], Cos[u]}
f1 = Show[{ParametricPlot[C1[u], {u, 0, 2 π}, PlotStyle → Red],
   ParametricPlot[C2[u], {u, 0, 2 π}, PlotStyle → Blue]}, PlotRange → All]
```

Out[205]=



In[206]:=
```
LoftedSurface[u_, v_] = {(1 - v) C2[u] + v C1[u], v}
f2 = ParametricPlot3D[LoftedSurface[u, v], {u, 0, 2 π}, {v, 0, 1}, Mesh → 4]
```

Out[206]= {{(2 - 0.1 u) (1 - v) Sin[u] + v Sin[u], (1 - v) Cos[u] + v Cos[u]}, v}

Out[207]=



In[208]:=
```
{{(2 - 0.1` u) (1 - v) Sin[u] + v Sin[u], (1 - v) Cos[u] + v Cos[u]}, v}
```

Out[208]= {{(2 - 0.1 u) (1 - v) Sin[u] + v Sin[u], (1 - v) Cos[u] + v Cos[u]}, v}

c) Extrusion Surface

In[209]:=
```
ClearAll["Global`*"]
C1[u_] := {Sin[u], Cos[u], 0}
v = {0, 2, 3};
f1 = ParametricPlot3D[C1[u], {u, 0, 2 π}, PlotStyle → Blue];
f2 = Graphics3D[{Blue, Arrow[Tube[{{0, 0, 0}, v}]]}];
Show[{f1, f2}, PlotRange -> All]
```

Out[214]=



In[215]:=
```
ExtrudeS[s_, t_] := C1[s] + v t
f3 =
  ParametricPlot3D[ExtrudeS[s, t], {s, 0, 2 π}, {t, 0, 1}, AspectRatio → Automatic,
   PlotStyle → { Specularity[White, 50], Opacity[0.7]}, Mesh → 4];
Show[{f1, f2, f3}, PlotRange -> All]
```

Out[217]=



d) Surface of Revolution

x-axis

In[218]:=
```
ClearAll["Global`*"]
C1[u_] := {2, 0} + {Sin[u], Cos[u]}
ParametricPlot[C1[u], {u, -π / 2, π / 2}, PlotStyle → Blue, AxesOrigin → {0, 0}]
m = Graphics3D[{PointSize[Large], Red, Point[{2, 0, 0}]}];
```

Out[220]=



In[222]:=
```
gx[u_] = C1[u][[1]];
gy[u_] = C1[u][[2]];
RevolveX[u_, θ_] := {gx[u], gy[u] Cos[θ], gy[u] Sin[θ]}
f1 = ParametricPlot3D[RevolveX[u, θ],
   {u, -π / 2, π / 2}, {θ, 0, 2 π}, AspectRatio → Automatic, Mesh → 4,
   PlotStyle → { Red, Specularity[White, 50], Opacity[0.3]}];
Show[{f1, m}, PlotRange → All, AxesLabel → {x, y, z}]
```

Out[226]=



y - axis

In[227]:=
```
C1[u_] := {2, 0} + {Sin[u], Cos[u]}
ParametricPlot[C1[u], {u, 0, 2 π}, PlotStyle → Blue, AxesOrigin → {0, 0}]
```

Out[228]=



In[229]:=
```
gx[u_] = C1[u][[1]];
gy[u_] = C1[u][[2]];
RevolveY[u_, θ_] := {gx[u] Cos[θ], gy[u], gx[u] Sin[θ]}
f2 = ParametricPlot3D[RevolveY[u, θ],
   {u, 0, 2 π}, {θ, 0, 2 π}, AspectRatio → Automatic, Mesh → 4,
   PlotStyle → { White, Specularity[White, 50], Opacity[0.3]}];
Show[{f1, f2, m}, PlotRange → All, AxesLabel → {x, y, z}]
```

Out[233]=



z - axis

In[234]:=
```
gx[u_] = C1[u][[1]];
gy[u_] = C1[u][[2]];
RevolveZ[u_, θ_] := {gx[u] Cos[θ], gx[u] Sin[θ], 0}
f3 = ParametricPlot3D[RevolveZ[u, θ], {u, 0, 2 π}, {θ, 0, 2 π},
  AspectRatio → Automatic, Mesh → 4, PlotStyle → { Green, Opacity[0.1]}]
```

Out[237]=



In[238]:=
```
Show[{f1, f2, f3, m}, PlotRange → All, AxesLabel → {x, y, z}]
```

Out[238]=



4) Tensor Product

Bezier Surface

In[239]:=
```
pts = {{{0, 0, 0}, {0, 1, 0}, {0, 2, 0}, {0, 3, 0}},
    {{1, 0, 0}, {1, 1, 1}, {1, 2, 1}, {1, 3, 0}},
    {{2, 0, 0}, {2, 1, 2}, {2, 2, 2}, {2, 3, 0}},
    {{3, 0, 0}, {3, 1, 1}, {3, 2, 1}, {3, 3, 0}},
    {{4, 0, 0}, {4, 1, 0}, {4, 2, 0}, {4, 3, 0}}};
f = BezierFunction[pts]
Show[Graphics3D[{PointSize[Medium], Red, Map[Point, pts]}],
 Graphics3D[{Gray, Line[pts], Line[Transpose[pts]]}],
 ParametricPlot3D[f[u, v], {u, 0, 1}, {v, 0, 1}, Mesh → True]]
```

Out[240]= BezierFunction[Argument count: 2 Output dimension: 3]

Out[241]=



B-Spline Surface

In[242]:=
```
b = Graphics3D[BSplineSurface[pts, SplineDegree → 2]];
Show[Graphics3D[{PointSize[Medium], Red,
   Map[Point, pts], Gray, Line[pts], Line[Transpose[pts]]}], b]
```

Out[243]=



NURBS

In[596]:=
```
pts = {{{0.5, 0, -0.5}, {0, 0, -0.5}, {0, 1, -0.5},
    {0.5, 1, -0.5}, {1, 1, -0.5}, {1, 0, -0.5}, {0.5, 0, -0.5}},
  {{0.5, 0, 0.7}, {0, 0, 0.7}, {0, 1, 0.7}, {0.5, 1, 0.7},
    {1, 1, 0.7}, {1, 0, 0.7}, {0.5, 0, 0.7}},
  {{0.5, 0, 0.9}, {0, 0, 0.9}, {0, 1, 1.5}, {0.5, 1, 1.5},
    {1, 1, 1.5}, {1, 0, 0.9}, {0.5, 0, 0.9}},
  {{0.5, -0.1, 1}, {0, -0.1, 1}, {0, 0.5, 2}, {0.5, 0.5, 2},
    {1, 0.5, 2}, {1, -0.1, 1}, {0.5, -0.1, 1}},
  {{0.5, -0.3, 1}, {0, -0.3, 1}, {0, -0.3, 2}, {0.5, -0.3, 2},
    {1, -0.3, 2}, {1, -0.3, 1}, {0.5, -0.3, 1}},
  {{0.5, -1.5, 1}, {0, -1.5, 1}, {0, -1.5, 2}, {0.5, -1.5, 2},
    {1, -1.5, 2}, {1, -1.5, 1}, {0.5, -1.5, 1}}};
w = {{1, .5, .5, 1, .5, .5, 1}, {1, .5, .5, 1, .5, .5, 1}, {1, .5, .5, 1, .5, .5, 1},
   {1, .5, .5, 1, .5, .5, 1}, {1, .5, .5, 1, .5, .5, 1}, {1, .5, .5, 1, .5, .5, 1}};
uk = {0, 0, 0, 1 / 4, 1 / 2, 3 / 4, 1, 1, 1};
vk = {0, 0, 0, 1 / 4, 1 / 2, 1 / 2, 3 / 4, 1, 1, 1};
```

In[247]:= `pts // MatrixForm`

Out[247]//MatrixForm=

$$\begin{pmatrix}
\begin{pmatrix}0.5\\0\\-0.5\end{pmatrix} & \begin{pmatrix}0\\0\\-0.5\end{pmatrix} & \begin{pmatrix}0\\1\\-0.5\end{pmatrix} & \begin{pmatrix}0.5\\1\\-0.5\end{pmatrix} & \begin{pmatrix}1\\1\\-0.5\end{pmatrix} & \begin{pmatrix}1\\0\\-0.5\end{pmatrix} & \begin{pmatrix}0.5\\0\\-0.5\end{pmatrix} \\
\begin{pmatrix}0.5\\0\\0.7\end{pmatrix} & \begin{pmatrix}0\\0\\0.7\end{pmatrix} & \begin{pmatrix}0\\1\\0.7\end{pmatrix} & \begin{pmatrix}0.5\\1\\0.7\end{pmatrix} & \begin{pmatrix}1\\1\\0.7\end{pmatrix} & \begin{pmatrix}1\\0\\0.7\end{pmatrix} & \begin{pmatrix}0.5\\0\\0.7\end{pmatrix} \\
\begin{pmatrix}0.5\\0\\0.9\end{pmatrix} & \begin{pmatrix}0\\0\\0.9\end{pmatrix} & \begin{pmatrix}0\\1\\1.5\end{pmatrix} & \begin{pmatrix}0.5\\1\\1.5\end{pmatrix} & \begin{pmatrix}1\\1\\1.5\end{pmatrix} & \begin{pmatrix}1\\0\\0.9\end{pmatrix} & \begin{pmatrix}0.5\\0\\0.9\end{pmatrix} \\
\begin{pmatrix}0.5\\-0.1\\1\end{pmatrix} & \begin{pmatrix}0\\-0.1\\1\end{pmatrix} & \begin{pmatrix}0\\0.5\\2\end{pmatrix} & \begin{pmatrix}0.5\\0.5\\2\end{pmatrix} & \begin{pmatrix}1\\0.5\\2\end{pmatrix} & \begin{pmatrix}1\\-0.1\\1\end{pmatrix} & \begin{pmatrix}0.5\\-0.1\\1\end{pmatrix} \\
\begin{pmatrix}0.5\\-0.3\\1\end{pmatrix} & \begin{pmatrix}0\\-0.3\\1\end{pmatrix} & \begin{pmatrix}0\\-0.3\\2\end{pmatrix} & \begin{pmatrix}0.5\\-0.3\\2\end{pmatrix} & \begin{pmatrix}1\\-0.3\\2\end{pmatrix} & \begin{pmatrix}1\\-0.3\\1\end{pmatrix} & \begin{pmatrix}0.5\\-0.3\\1\end{pmatrix} \\
\begin{pmatrix}0.5\\-1.5\\1\end{pmatrix} & \begin{pmatrix}0\\-1.5\\1\end{pmatrix} & \begin{pmatrix}0\\-1.5\\2\end{pmatrix} & \begin{pmatrix}0.5\\-1.5\\2\end{pmatrix} & \begin{pmatrix}1\\-1.5\\2\end{pmatrix} & \begin{pmatrix}1\\-1.5\\1\end{pmatrix} & \begin{pmatrix}0.5\\-1.5\\1\end{pmatrix}
\end{pmatrix}$$

```
In[599]:= a = Graphics3D[{
      FaceForm[Yellow, Blue],
      BSplineSurface[pts, SplineKnots → {uk, vk}, SplineDegree → 2, SplineWeights → w,
       SplineClosed → {False, True}]}, ViewPoint → {Right, Front}, Boxed → False];
   Show[Graphics3D[{PointSize[Medium], Red, Map[Point, pts],
      Gray, Line[pts], Line[Transpose[pts]],
      Text[Style["b00", Red, Italic, 24], pts[[1, 1]]],
      Text[Style["b01", Red, Italic, 24], pts[[1, 2]]],
      Text[Style["b02", Red, Italic, 24], pts[[1, 3]]],
      Text[Style["b04", Red, Italic, 24], pts[[1, 4]]],
      Text[Style["b05", Red, Italic, 24], pts[[1, 5]]],
      Text[Style["b06", Red, Italic, 24], pts[[1, 6]]],
      Text[Style["b07", Red, Italic, 24], pts[[1, 7]]],

      (*Text[Style["b10",Red,Italic,24],pts[[2,1]]],
      Text[Style["b11",Red,Italic,24],pts[[2,2]]],
      Text[Style["b12",Red,Italic,24],pts[[2,3]]],
      Text[Style["b14",Red,Italic,24],pts[[2,4]]],
      Text[Style["b15",Red,Italic,24],pts[[2,5]]],
      Text[Style["b16",Red,Italic,24],pts[[2,6]]],
      Text[Style["b17",Red,Italic,24],pts[[2,7]]],*)

      Text[Style["b50", Red, Italic, 24], pts[[6, 1]]],
      Text[Style["b51", Red, Italic, 24], pts[[6, 2]]],
      Text[Style["b52", Red, Italic, 24], pts[[6, 3]]],
      Text[Style["b54", Red, Italic, 24], pts[[6, 4]]],
      Text[Style["b55", Red, Italic, 24], pts[[6, 5]]],
      Text[Style["b56", Red, Italic, 24], pts[[6, 6]]],
      Text[Style["b57", Red, Italic, 24], pts[[6, 7]]]
     }], a]
```

Out[600]=



In[602]:=
```
Manipulate[
 Show[Graphics3D[{PointSize[Medium], Red, Map[Point, pts],
    Gray, Line[pts], Line[Transpose[pts]]}], Graphics3D[{
    FaceForm[Yellow, Blue],
    BSplineSurface[pts, SplineKnots → {uk, vk}, SplineDegree → 2, SplineWeights →
       {{1, a1, a2, a3, a4, .5, 1}, {1, a1, a2, a3, a4, .5, 1}, {1, a1, a2, a3, a4, .5, 1},
        {1, a1, a2, a3, a4, .5, 1}, {1, a1, a2, a3, a4, .5, 1}, {1, a1, a2, a3, a4, .5, 1}},
     SplineClosed → {False, True}]}], ViewPoint → {Right, Front}, Boxed → False]],
 {a1, 0.1, 1}, {a2, 0.1, 1}, {a3, 0.1, 1}, {a4, 0.1, 1}]
```

Out[602]=